

Projet de compilation

27 mars 2018 – Version corrigée qui remplace celle du 11 mars 2018

Le projet consiste à analyser ce langage impératif qui ressemble à Pascal, à l'interpréter, à le traduire en code à trois adresses et enfin à émuler une machine qui fait tourner ce code.

1 Langage source : définition de la syntaxe

Le langage est inspiré du langage de programmation Pascal¹. Nous l'appellerons « *Léa* » (Langage élémentaire algorithmique).

Le langage *Léa* est défini par la grammaire suivante :

Programme

```
1 program:
2   type_declaration_part
3   variable_declaration_part
4   procedure_and_function_definition_part
5   TOKEN_BEGIN
6   statement_list
7   TOKEN_END
```

Déclaration des types nommés

```
1 type_declaration_part:
2   'type' type_declaration_list
3   | /* empty */
4
5 type_declaration_list:
6   type_declaration_list type_declaration
7   | type_declaration
8
9 type_declaration:
10  IDENTIFIER '=' type ';' ;
```

1. <http://www.fit.vutbr.cz/study/courses/APR/public/ebnf.html>

Définition des types

```
1 type:
2   simple_type
3   | named_type
4   | index_type
5   | array_type
6   | pointer_type
7
8 simple_type:
9   'character'
10  | 'integer'
11  | 'boolean'
12
13 named_type:
14   IDENTIFIER
15
16 index_type:
17   enumerated_type
18   | subrange_type
19
20 enumerated_type:
21   '(' identifier_list ')'
22
23 subrange_type:
24   INTEGER '..' INTEGER
25
26 array_type:
27   'array' '[' range_type ']' 'of' type
28
29 range_type:
30   enumerated_type
31   | subrange_type
32   | named_type
33
34 pointer_type:
35   '^' type
```

Déclaration des variables typées

```
1 variable_declaration_part:
2   'var' variable_declaration_list
3   | /* empty */
4
5 variable_declaration_list:
6   variable_declaration_list variable_declaration
7   | variable_declaration
8
9 variable_declaration:
10  identifier_list ':' type ';'
11
```

```

11
12 identifier_list:
13     identifier_list ',' IDENTIFIER
14     | IDENTIFIER

```

Déclaration et définition des procédures et des fonctions

```

1 procedure_and_function_definition_part:
2     procedure_and_function_definition_list
3     | /* empty */
4
5 procedure_and_function_definition_list:
6     procedure_and_function_definition_list
7     procedure_and_function_definition
8     | procedure_and_function_definition
9
10 procedure_and_function_definition:
11     procedure_and_function_definition_head
12     variable_declaration_part block
13     | procedure_and_function_definition_head ';'
14
15 procedure_and_function_definition_head:
16     'procedure' IDENTIFIER '(' argt_part ')'
17     | 'function' IDENTIFIER '(' argt_part ')' ':' type
18
19 argt_part:
20     argt_list
21     | /* empty */
22
23 argt_list:
24     argt_list ',' argt
25     | argt
26
27 argt:
28     IDENTIFIER ':' type

```

Blocs

```

1 block:
2     'begin'
3     statement_list
4     'end'

```

Instructions

```

1 statement_list:
2     statement_list statement

```

```

3 | statement
4
5 statement:
6     simple_statement
7     | structured_statement
8
9 simple_statement:
10    assignment_statement
11    | procedure_statement
12    | new_statement
13    | dispose_statement
14    | println_statement
15    | readln_statement
16    | return_statement
17
18 assignment_statement:
19     variable_access ':' '=' expression ';'
20
21 procedure_statement:
22     procedure_expression ';'
23
24 procedure_expression:
25     IDENTIFIER '(' expression_part ')'
26
27 expression_part:
28     expression_list
29     | /* empty */
30
31 expression_list:
32     expression_list ',' expression
33     | expression
34
35 new_statement:
36     'new' '(' variable_access ')' ';'
37
38 dispose_statement:
39     'dispose' '(' variable_access ')' ';'
40
41 println_statement:
42     'println' '(' expression ')' ';'
43
44 readln_statement:
45     'readln' '(' expression ')' ';'
46
47 return_statement:
48     'return' '(' expression ')' ';'
49
50 structured_statement:
51     block
52     | if_statement
53     | while_statement

```

```

54
55 if_statement:
56     'if' expression 'then' statement
57     | 'if' expression 'then' statement 'else' statement
58
59 while_statement:
60     'while' expression 'do' statement

```

Expressions

```

1  variable_access
2      : IDENTIFIER
3      | indexed_variable
4      | variable_access '^'
5
6  indexed_variable:
7      variable_access '[' expression ']'
8
9  expression:
10     variable_access
11     | expression '+' expression
12     | expression '*' expression
13     | expression '-' expression
14     | expression '/' expression
15     | expression '||' expression
16     | expression '&&' expression
17     | '-' expression
18     | '!' expression
19     | expression '<' expression
20     | expression '<=' expression
21     | expression '>' expression
22     | expression '>=' expression
23     | expression '=' expression
24     | expression '!=' expression
25     | '(' expression ')'
26     | procedure_expression
27     | literal

```

Expressions littérales

```

1  literal:
2      INTEGER
3      | CHARACTER
4      | 'true'
5      | 'false'
6      | 'null'

```

Où les symboles INTEGER, CHARACTER et IDENTIFIER représentent les unités lexicales suivantes :

INTEGER Une constante numérique entière de format du langage C. Exemples : -32768, 32767

CHARACTER Une constante représentant un caractère. Exemples : 'a', '\'', '\\'

IDENTIFIER Identificateur au format du langage C.

Toutes les autres unités lexicales sont représentées littéralement dans la grammaire entre apostrophes.

2 Langage source : système de type

Une expression de type est telle que

- `character`, `integer`, `boolean` sont des types
- si τ est un type, alors `pointer(τ)` est un type
- si τ est un type, alors `array(τ , min , max)` est un type, min , max des constantes entières (définissant l'intervalle du tableau)
- si τ_1 et τ_2 sont des types, alors $\tau_1 \times \tau_2$ est un type (type produit des arguments d'une fonction ou d'une procédure)
- si τ_1 et τ_2 sont des types, alors $\tau_1 \rightarrow \tau_2$ est un type (type fonction)

Bonne formation d'un code écrit en langage Léa

Un programme écrit en *Léa* est un texte engendré par cette grammaire qui vérifie les contraintes suivantes :

- Tous les identificateurs (types définis, variables, fonctions et procédures) sont déclarés avant d'être utilisés. Les fonctions et procédures peuvent être déclarées puis ensuite définies, ou seulement définies.
- Toute expression est bien typée, c'est-à-dire qu'il lui correspond une expression de type telle que définie plus haut et selon les tableaux suivants :

Opérateur binaire

	Opérande 1	Opérande 2	Valeur
<code>+</code> , <code>*</code> , <code>-</code> , <code>/</code>	<code>integer</code>	<code>integer</code>	<code>integer</code>
<code><</code> , <code>></code> , <code><=</code> , <code>>=</code>	<code>integer</code> <code>character</code>	<code>integer</code> <code>character</code>	<code>boolean</code>
<code>=</code> , <code>!=</code>	<code>integer</code> <code>character</code> <code>pointer(τ)</code>	<code>integer</code> <code>character</code> <code>pointer(τ)</code>	<code>boolean</code>
<code>&&</code> , <code> </code>	<code>boolean</code>	<code>boolean</code>	<code>boolean</code>

Opérateur unaire

	Opérande	Valeur
<code>-</code>	<code>integer</code>	<code>integer</code>
<code>!</code>	<code>boolean</code>	<code>boolean</code>

Expression ²

	Valeur
<code>true, false</code>	<code>boolean</code>
<code>null</code>	<i>pointer</i> (τ)
<code>INTEGER</code>	<code>integer</code>
<code>CHARACTER</code>	<code>character</code>
<code>IDENTIFIER</code>	type de la variable
<code>variable_access</code> [$e : integer$]	type τ_1 si <code>variable_access</code> est de type $array(\tau_1, min, max)$
<code>IDENTIFIER</code> ($a_1 : \tau_1, a_2 : \tau_2, \dots$)	type μ si le type de la fonction est $(\tau_1 \times (\tau_2 \times \dots)) \rightarrow \mu$
<code>variable_access</code> '^'	type τ si <code>variable_access</code> est de type <i>pointer</i> (τ)

- Toute instruction est bien typée :
 - L'affectation doit avoir des membres gauche et droit de même type.
 - Le test et la boucle ont comme argument une expression de type booléen.
 - L'instruction `return` a comme argument une expression de même type que le type de retour de la fonction qui la contient.
 - Les instruction `new` et `dispose` ont comme argument un pointeur.

2. On remarque que la bonne formation du typage statique ne suffit pas à vérifier les indices possibles des tableaux.

3 Langage source : définition de la sémantique

Idée générale

La sémantique définit l'évolution de l'état courant d'une machine abstraite sous l'effet d'une instruction (ou d'une expression qui a un effet) de *Léa*.

Un état Q est un triplet (G, S, H) où G (*Global*) est un environnement statique, S (*Stack*) est une pile et H (*Heap*) un tas.

Une instruction s fait passer la machine d'un état $Q = (G, S, H)$ à un état $Q' = (G', S', H')$, ce que l'on note

$$Q/s \Rightarrow Q'$$

Une expression e fait passer la machine d'un état Q à un état Q' , et produit la valeur v , ce que l'on note

$$Q/e \Rightarrow Q'/v$$

Environnements

Un environnement E est une fonction qui associe une valeur $E(x)$ à un identificateur x . Le domaine d'un environnement est l'ensemble des variables qu'il définit.

Environnement global ou *statique*

L'environnement global G a un domaine qui reste inchangé pendant toute la durée de l'exécution du programme.

Enregistrement d'activation

Un enregistrement d'activation est l'ensemble des informations relatives à l'exécution d'une procédure (resp. d'une fonction). Pour représenter un enregistrement d'activation, nous utiliserons un environnement *local* dans lequel nous distinguerons une variable $\#return\#$ pour la valeur de retour d'une fonction, et $\#save\#$ pour l'état de la machine avant que la procédure (resp. la fonction) soit appelée. Les arguments et variables locales à la procédure (resp. à la fonction) seront stockés dans l'environnement local.

L'enregistrement d'activation est empilé dans la pile S lors de l'appel d'une procédure ou d'une fonction, il est dépilé lorsque le contrôle retourne à l'appelant.

La valeur d'une variable x est $E(x)$ où E désigne Top si $x \in dom(Top)$ et G si $x \in dom(G) - dom(Top)$. Top désigne le sommet de la pile S et G l'environnement global.

Tas

Le tas H est une fonction partielle qui à chaque adresse $a \in \mathbb{Z}/2^{32}\mathbb{Z}$ pour laquelle $H(a)$ est définie, associe une valeur dans $\mathbb{Z}/2^8\mathbb{Z}$.

C'est-à-dire qu'à chaque adresse définie correspond un octet.

Le tas est modifié explicitement selon deux fonctions **new()** (allocation de mémoire dynamique) et **dispose()** (libération de la mémoire dynamique) prévues dans le langage *Léa*.

new(x) a comme effet d'allouer $taille(\tau)$ octets dans le tas où τ est le type de x , et d'affecter à x l'adresse du premier octet. La prochaine adresse où une nouvelle allocation est possible est augmentée de $taille(\tau)$.

dispose(x) a comme effet de désallouer les octets réservés à x , et d'affecter à x la valeur **null**. Aucun ramasse-miettes n'est prévu dans l'implémentation de ce projet.

La valeur pointée par une variable x dont le type est $pointeur(\tau)$ est un ensemble d'octets (dépendant du type τ) commençant à $H(E(x))$ où E est l'environnement de x .

La seule façon de connaître une adresse en langage *Léa*, est d'utiliser un pointeur p qui la désigne. Une adresse non encore allouée est égale à **null**.

Valeurs des expressions typées

Aux expressions de type correspondent un ensemble de valeurs dans $\mathbb{Z}/n\mathbb{Z}$.

Le tableau suivant donne pour chaque type, le nombre d'octets alloués en mémoire et l'encodage attendu :

Type	Image	Taille (en octets)	Encodage
integer	$\mathbb{Z}/2^{32}\mathbb{Z}$	2	encodés comme le sont les <i>signed int</i> en C (1 bit pour le signe, 31 pour le nombre). C'est-à-dire des relatifs compris dans l'intervalle $[-2^{32-1}, 2^{32-1} - 1]$.
character	$\mathbb{Z}/2^8\mathbb{Z}$	1	ISO 8859-15
boolean	$\mathbb{Z}/2^8\mathbb{Z}$	1	false si = 0, true si $\neq 0$
$pointeur(\tau)$	$\mathbb{Z}/2^{32}\mathbb{Z}$	4	Sans objet
$array(\tau, min, max)$	–	$k \times taille(\tau)$	Sans objet

Valeurs par défaut des expressions et instructions

Chaque expression de type τ a une valeur par défaut selon le tableau suivant

type	valeur par défaut
integer	0
character	'\0'
boolean	FALSE
$pointeur(\tau)$	null
$array(\tau, min, max)$	valeurs par défaut de τ sur tout le tableau

Une fonction a comme valeur par défaut la valeur par défaut de son type de sortie. L'absence d'instruction *return(x)* ne provoque pas d'erreur d'exécution.

Sémantique des opérateurs

La sémantique des opérateurs arithmétiques sur les entiers et les réels est une application $(\mathbb{Z}/n\mathbb{Z})^k \rightarrow \mathbb{Z}/n\mathbb{Z}$, où k est l'arité de l'opérateur.

La sémantique des opérateurs logiques sur les booléens est une application $(\mathbb{Z}/2\mathbb{Z})^k \rightarrow \mathbb{Z}/2\mathbb{Z}$, où k est l'arité de l'opérateur.

La sémantique des opérateurs de comparaison est une application $(\mathbb{Z}/n\mathbb{Z})^2 \rightarrow \mathbb{Z}/2\mathbb{Z}$.

Tous ces opérateurs ont une sémantique habituelle ; celle que l'on trouve dans le langage C. La sémantique de l'opérateur *op* sera notée $\llbracket op \rrbracket$. Exemple : $\llbracket 2 + 3 \rrbracket = \llbracket + \rrbracket(2, 3)$

Sémantique des expressions et instructions

La procédure `println(x)` permet de provoquer un effet vers la sortie standard. L'affichage sera alors la valeur de l'expression *x* selon son type, suivi du caractère '`\n`'. Les valeurs sont

- Entier relatif décimal signé pour `integer`,
- *true* ou *false* pour `boolean`,
- Le caractère ISO 8859-15 pour `character`,
- `ARRAY` pour $array(\tau, min, max)$,
- `POINTER` pour $pointer(\tau)$.

La procédure `readln(x)` lit depuis l'entrée standard une valeur entière. *x* doit être de type `integer`, `character` ou `boolean` et encodé comme vu page 10.

Si *p* est une variables de type $pointer(\tau)$, l'expression \hat{p} désigne la valeur pointée à l'adresse *p*. Si *p* a comme valeur *null*, le programme produit une erreur système.

Si *t* est une variables de type $array(\tau, min, max)$, l'expression $t[e]$ désigne la valeur de $t + (\sigma(e) - min)$ où $\sigma(e)$ est la valeur entière de l'expression *e*. Si $i < min$ ou $i > max$, la valeur est indéterminée et le programme produit une erreur système.

Les procédures et fonctions utilisent exclusivement un passage de paramètres par valeur et les valeurs ne peuvent être que de type simple³.

Les affectations ne peuvent se faire que sur les expressions dites *accessibles*, c'est-à-dire, les variables de type simple, les expressions $t[e]$ où *t* est de type $array(\tau, min, max)$, les expressions \hat{p} où *p* est de type $pointer(\tau)$.

3. Les types simples sont `integer`, `boolean` et `character`

4 Langage C3A

Le langage utilisé lors de la compilation est un langage à trois adresses qui consiste en une liste d'instructions élémentaires écrites sous forme de tuples :

(Étiquette, Instruction, Opérande₁, Opérande₂, Opérande₃)

- **Étiquette** est le numéro du tuple.
- **Instruction** est le numéro de l'instruction (de 1 à 17).
- **Opérande_i** : Chaque instruction a zéro, une, deux ou trois opérandes. Un opérande est soit une étiquette, soit un nom de registre r_i , soit un numéral sur $[-2^{32-1}, 2^{32-1} - 1]$.

Chacune des instructions a un effet sur une machine abstraite (qui est une machine de Von Neumann) qui possède les propriétés suivantes :

- La machine possède un tableau d'instructions. Ce tableau est le programme en cours d'exécution.
- La machine a une infinité de registres r_i où sont stockées des valeurs sur $[-2^{32-1}, 2^{32-1} - 1]$.
- Un registre est distingué pour y mettre le numéro de l'instruction en cours. Ce registre est le *registre d'instruction*, sa valeur est zéro initialement.
- La machine a trois espaces mémoire T , S et H . Ce sont des tableaux de 2^{32} octets⁴, T est réservé aux données statiques, S est la pile et H le tas. L'organisation de la pile et du tas est une liste, simplement chaînée pour la pile, doublement chaînée pour le tas. Comme il n'est jamais prévu d'insérer un élément au milieu de la pile ou du tas, mais seulement à leur extrémité, et que chaque élément de ces structures est un tableau d'octets, l'encodage de T , S et H est un tableau d'octets. On remarque que le *mot* de la machine est une séquence de 4 octets. Ceci veut dire que tout espace réservé pour une valeur est 4 octets, y compris pour les booléens et les caractères.
- Chaque instruction produit un calcul élémentaire ; produit une affectation de la mémoire ou une modification d'un registre. Chacune de ces opérations se fait invariablement sur des **mots** de 4 octets.
- La machine prend en entrée et produit en sortie un flux de caractères encodés en ISO 8859-15 par l'intermédiaire de deux routines *read* et *print*. Ces routines seront simplement codées grâce aux commandes `printf` et `scanf` du langage C.

Les instructions sont les suivantes :

1. **skip**
Ne fait rien.
2. **stop**
Arrête la machine.
3. **x := y bop z**
Instruction d'affectation où *bop* est un opérateur binaire arithmétique (+, −, ×, /), logique (∧, ∨), ou de comparaison (<, >, ≤, ≥, =, ≠).

4. Une variante possible est la construction d'une mémoire qui consiste en tableaux de *wint32_t*.

4. **x := uop y**
Instruction d'affectation où *uop* est un opérateur unaire arithmétique (−) ou logique (¬)
5. **x := y**
Instruction de copie
6. **goto x**
Branchement inconditionnel ; *x* est une étiquette
7. **if x goto y**
Branchement conditionnel. Cette instruction a pour effet de faire exécuter l'instruction étiquetée par *y* si la valeur de *x* est différente de zéro, l'instruction suivante dans le cas contraire.
8. **push x**
Empile un nouvel enregistrement d'activation de *x* octets dans *S*. Cela a pour effet de créer un nouvel espace mémoire de *x* octets qui sera le prochain enregistrement d'activation au moment du **call**.
9. **param x**
Enregistre la valeur *x* dans la liste des arguments du nouvel enregistrement d'activation.
10. **call x**
— Le nouvel enregistrement d'activation devient l'actuel.
— Sauvegarde la valeur du registre d'instructions dans l'enregistrement actuel.
— Affecte au registre d'instruction la valeur *x*.
Ceci a pour effet que la prochaine instruction exécutée est celle étiquetée par *x*.
11. **return x**
— Enregistre *x* dans la valeur de retour de l'enregistrement d'activation actuel.
— Affecte au registre d'instruction la valeur sauvegardée dans l'enregistrement actuel.
— Dépille *S*.
L'enregistrement d'activation actuel devient celui qui était pointé par **previous** au sommet de la pile.
12. **x := T[y]**
Affecte à *x* la valeur de la mémoire située à *y* octets au-delà de *T*. *T* est soit *S*, soit l'enregistrement d'activation actuel (en sommet de pile), soit *H*.
13. **T[x] := y**
Affecte *y* dans la mémoire située à *x* octets au-delà de *y*. *T* est soit *S*, soit l'enregistrement d'activation actuel, soit *H*.
14. **x := malloc y**
Affecte à *x* la valeur du contenu de l'adresse d'une zone de *y* octets allouée dans le tas *H*.
15. **free x**
Libère la zone *x* allouée dans le tas *H*.
16. **print x y**
Affiche en sortie standard la valeur *x*. *y* est un nombre désignant le type de *x*.
— ISO 8859-15 sur l'octet de poids faible si *y* vaut 0. **Étiquette** est le numéro du tuple.

- Un numérique sur $[-2^{32-1}, 2^{32-1} - 1]$ si y vaut 1 ;
- `true` ou `false` selon la valeur de x si y vaut 2.

17. `x := read`

Lit l'entrée standard pour encoder une valeur numérale sur $[-2^{32-1}, 2^{32-1} - 1]$ et l'affecte à x .

4.1 Effet du code *C3A* sur la machine

La machine lit l'instruction dont le numéro se trouve dans le registre d'instruction et l'exécute. Toutes les instructions à l'exception de `goto`, `if x goto y`, `call x` et `return x` sont suivies d'une instruction qui incrémente le compteur d'instruction. Le processus se réitère jusqu'à l'exécution de l'instruction `stop`.

5 Travail à réaliser

1. Un analyseur lexical du langage Léa. Outre que cet analyseur doit reconnaître l'ensemble des tokens utilisés pour l'analyse syntaxique, il devra aussi reconnaître et encoder correctement :
 - Les commentaires de type C (`/* */`)
 - Les commentaires de type C++ (`//`)
 - Les littéraux `integer` codés en décimal et en hexadécimal (*i.e.* `0x1CFE`)
 - Les littéraux `character` codés en décimal, octal, hexadécimal et écrits entre apostrophes.Cet analyseur lexical conservera le numéro de ligne, le numéro de colonne pour d'éventuels messages d'erreur.
2. Un analyseur syntaxique pour le langage Léa.

Cette fonction teste que le texte est bien engendré par la grammaire de Léa et envoie un message d'erreur précis dans le cas contraire.

Elle construit une table de symboles pour
 - Les définitions de types
 - Les variables globales
 - Les fonctions et les procédures
 - Les variables locales
3. Un analyseur sémantique pour Léa

Chaque expression doit avoir un type bien formé (selon la section ??).⁵

L'analyseur sémantique envoie un message explicite en cas de mal formation, et construit le typage des expressions dans le cas contraire.
4. Un interpréteur du langage Léa. Il s'agit d'une fonction qui prend en entrée un texte en Léa, un environnement global, un tas, une pile (G, H, S) et retourne un nouvel état (G', H', S') en produisant un effet. Les seuls effets possibles sont
 - Sortie vers la sortie erreur standard (`stderr`) d'un message d'erreur le cas échéant
 - Sortie vers la sortie standard (`stdout`) des affichages éventuels avec la procédure `println()`
5. Un compilateur de Léa vers C3A.⁶
6. Un émulateur C3A. Il s'agit d'une fonction qui prend en entrée un programme C3A, un environnement global, un tas, une pile (G, H, S) et retourne un nouvel état (G', H', S') en produisant un effet.⁷

5. La bonne formation du type dépend de la nature et du contexte de l'expression (*i.e.* comme opérande de type τ).

6. Les enseignants fourniront une bibliothèque où les structures de données pour C3A seront déjà réalisées.

7. Les enseignants fourniront un prototype de programme à compléter où l'analyse syntaxique du code C3A est déjà réalisé.

6 Modalités de réalisation

6.1 Éléments fournis par les enseignants

- Un analyseur syntaxique pour le langage C3A qui produit une structure de donnée de code à 3 adresses :
 - `emulator.y`
 - `emulator.l`
 - `c3a.h`
 - `c3a.c`
- Un fichier `Makefile` complet et suffisant pour l'ensemble du projet
- des fichiers d'exemple `progr-xxx.lea`, `progr-xxx.c3a` contenant des exemples de programmes en langage Léa ou C3A avec en commentaire les réalisations attendues.

Groupes Le projet doit être réalisé par des groupes de 4 étudiants⁸ (ou moins à la discrétion du chargé de TD). Les groupes sont constitués sur leur proposition par l'enseignant.

Chaque groupe est représenté par l'un des trois étudiants qui est le *correspondant*. Le *correspondant* n'a aucune autorité sur le reste du groupe, son rôle est :

- De correspondre avec l'enseignant
- De correspondre avec les trois autres étudiants
- D'administrer le dépôt *GIT* ou *SVN*

6.2 Livrable

Le livrable doit être déposé par le correspondant du groupe sur un dépôt du Cremi accessible par l'enseignant selon les modalités exigées lors des séances de TD et dès la première séance de TD de la semaine du 19 mars.

La dernière version devra être déposée avant le mardi 10 avril à 23h59 et ne plus être modifié après.

Le livrable contient les fichiers suivants :

- `README`
Qui contient en langue anglaise ou française les commandes nécessaires pour compiler le projet complet.
- `AUTHORS`
Qui contient le noms des 4 étudiants dans l'ordre alphabétique.
- `documentation.pdf`
Un rapport d'environ 10 pages qui explique le plan de votre programme, les difficultés rencontrées, les solutions apportées, les tests effectués. Ce document doit pouvoir être lu indépendamment des fichiers sources.
Si un ou plusieurs étudiants du groupe a des difficultés à rédiger correctement en langue française parce qu'il ne serait pas francophone, il lui est autorisé de rédiger en langue anglaise une section ou un chapitre entier. Dans tous les cas, la rédaction doit être soignée, tant dans l'expression que dans la forme (erreurs typographiques, mise en page, etc.).
- Les sources

8. dans ce qui suit *étudiant*, *correspondant*, etc. valent pour *étudiant.e*, *correspondant.e* etc.


```
src/Makefile
src/c3a.l
src/c3a.y
src/lea.l
src/lea.y
src/*.h
src/*.c
```

Les langages suivants sont utilisés :

- **Makefile** invoqué par la commande **make** pour construire les dépendances entre fichiers.
- **Bison** qui est une implémentation GNU du compilateur de compilateur **Yacc**.
- **Flex** qui est une implémentation BSD de l'analyseur lexical **Lex**.
- Langage **C** et bibliothèques standards ANSI-C (`<stdio.h>` , `<stdlib.h>`, etc.)

Les commandes suivantes doivent pouvoir être lancées depuis le répertoire source :

1. **make all**
Compile tous les exécutables ; efface les fichiers provisoires et lance tous les tests.
2. **make parser**
Compile l'analyseur syntaxique et l'analyseur sémantique **parser**.
3. **make interpreter**
Compile l'interpréteur du langage *Léa* **interpreter**
4. **make compiler**
Compile le compilateur du langage Léa **compiler**
5. **make emulator**
Compile l'émulateur du langage C3A **emulator**
6. **make <file>.err**
Étant donné un fichier source *Léa* `<file>.lea` et **parser**, lance la commande
`./parser < <file>.lea 2> <file>.err`
7. **make <file>.lea.out**
Étant donné un fichier source *Léa* `<file>.lea` et **interpreter**, lance la commande
`./interpreter < <file>.lea > <file>.lea.out`
8. **make <file>.c3a**
Étant donné un fichier source *Léa* `<file>.lea` et **compiler**, lance la commande
`./compiler < <file>.lea > <file>.c3a`
9. **make <file>.c3a.out**
Étant donné un fichier source *Léa* `<file>.c3a` et **emulator**, lance la commande
`./emulator < <file>.c3a > <file>.c3a.out`